# Maintainer: Laurent Carlier <lordheavym@gmail.com>
# Contributor: Jaroslav Lichtblau <dragonlord@aur.archlinux.org>
# Contributor: Allan McRae <allan@archlinux.org>
# Contributor: geekitus

pkgname=blobwars-data
_pkgname=blobwars
pkgver=2.00
pkgrel=3
pkgdesc="Platform action game featuring a blob wih lot of weapons - data files"
arch=('any')
url="https://sourceforge.net/projects/blobwars"
license=('GPL')
source=(https://downloads.sourceforge.net/project/${_pkgname}/${_pkgname}-${pkgver}.tar.gz)
sha512sums=('412afd029554e59e7947466f36b51ef863faf1712898397498d5497e9a38dced92777b7684c1fb231998d893a6ce2c435b8f4278429df41168385de8fa18725b')

package() {
  cd blobwars-${pkgver}

  install -dm755 "${pkgdir}"/usr/share/games/blobwars/
  cp -r {data,gfx,sound,music} "${pkgdir}"/usr/share/games/blobwars/
  # fix world readable bit
  find "${pkgdir}"/usr/share/games/blobwars/. -type d -exec chmod 755 {} +
  find "${pkgdir}"/usr/share/games/blobwars/. -type f -exec chmod 644 {} +
}
